

Haskell Quick Reference

COSC 8—Fall 2007

This document summarizes a number of common Haskell functions defined in the standard Prelude. This information is based on Chapter 23 of “The Haskell School of Expression.”

Simple List Selector Functions

The functions *head* and *tail* extract the first element and remaining elements, respectively, from a list, which must be nonempty. The functions *last* and *init* are the dual functions that work from the end of a list, rather than from the beginning. The *null* function tests to see if a list is empty.

```
head      :: [a] → a
head (x : _) = x
head _     = error "head: empty list"

last      :: [a] → a
last [x]   = x
last (_ : xs) = last xs
last []    = error "last: empty list"

tail      :: [a] → [a]
tail (_ : xs) = xs
tail []      = error "tail: empty list"
```

```
init      :: [a] → [a]
init [x]   = []
init (x : xs) = x : init xs
init []    = error "init: empty list"
```

```
null      :: [a] → Bool
null []    = True
null (_ : _) = False
```

Index-Based Selector Functions

To select the *m*th element from a list, with the first element being the 0th element, we can use the indexing function (*!!*). The value *take n xs* is the prefix of *xs* of length *n*, or *xs* itself if *n > length xs*. Similarly, *drop n xs* is the suffix of *xs* after the first *n* elements, or [] if *n > length xs*. Finally, *splitAt n xs* is equivalent to (*take n xs*, *drop n xs*).

```
(!!)      :: [a] → Int → a
(x : _) !! 0 = x
(_ : xs) !! n | n > 0 = xs !! (n - 1)
(_ : _) !! _ = error "!!: negative index"
[] !! _     = error "!!: index too large"
```

```
take      :: Int → [a] → [a]
take 0 _ = []
take _ [] = []
take n (x : xs) | n > 0 = x : take (n - 1) xs
take _ _ = error "take: negative argument"
```

```
drop      :: Int → [a] → [a]
drop 0 xs = xs
```

```
drop _ [] = []
drop n (_ : xs) | n > 0 = drop (n - 1) xs
drop _ _ = error "drop: negative argument"
```

```
splitAt   :: Int → [a] → ([a], [a])
splitAt 0 xs = ([], xs)
splitAt _ [] = ([], [])
splitAt n (x : xs) | n > 0 = (x : xs', xs'')
    where (xs', xs'') = splitAt (n - 1) xs
splitAt _ _ = error "splitAt: negative argument"
```

```
length    :: [a] → Int
length [] = 0
length (_ : xs) = 1 + length xs
```

Predicate-Based Selector Functions

The value *takeWhile p xs* is the longest (possibly empty) prefix of *xs*, all of whose elements satisfy the predicate *p*. The value *dropWhile p xs* is the remaining suffix.

```
takeWhile :: (a → Bool) → [a] → [a]
takeWhile p [] = []
takeWhile p (x : xs)
    | p x = x : takeWhile p xs
    | otherwise = []
```

```
dropWhile :: (a → Bool) → [a] → [a]
dropWhile p [] = []
dropWhile p xs@(x : xs')
    | p x = dropWhile p xs'
    | otherwise = xs
```

The function *span* *p xs* is equivalent to (*takeWhile p xs*, *dropWhile p xs*), while *break p* uses the negation of *p*. The function *filter* removes all elements of a list not satisfying a predicate.

```
span, break :: (a → Bool) → [a] → ([a], [a])
span p [] = ([], [])
span p xs@(x : xs')
  | p x = let (xr, xt) = span p xs' in (x : xr, xt)
  | otherwise = ([], xs)

break p = span (not ∘ p)
```

```
filter :: (a → Bool) → [a] → [a]
filter p [] = []
filter p (x : xs)
  | p x = x : filter p xs
  | otherwise = filter p xs
```

Fold-like Functions

The functions *foldl1* and *foldr1* are variants of *foldl* and *foldr* that have no starting value argument, and thus must be applied to nonempty lists.

```
foldl :: (a → b → a) → a → [b] → a
foldl f z [] = z
foldl f z (x : xs) = foldl f (f z x) xs
```

```
foldl1 :: (a → a → a) → [a] → a
foldl1 f (x : xs) = foldl f x xs
foldl1 _ [] = error "foldl1: empty list"
```

```
foldr :: (a → b → b) → b → [a] → b
foldr f z [] = z
foldr f z (x : xs) = f x (foldr f z xs)
```

```
foldr1 :: (a → a → a) → [a] → a
foldr1 f [x] = x
foldr1 f (x : xs) = f x (foldr1 f xs)
foldr1 _ [] = error "foldr1: empty list"
```

The function *scanl* is similar to *foldl*, but returns a list of successive reduced values from the left. The function *scanr* is the analogue for *foldr*.

```
scanl :: (a → b → a) → a → [b] → [a]
scanl f q xs = q : (case xs of
  [] → []
  (x : xs') → scanl f (f q x) xs')

scanl1 :: (a → a → a) → [a] → [a]
scanl1 f (x : xs) = scanl f x xs
scanl1 _ [] = error "scanl1: empty list"

scanr :: (a → b → b) → b → [a] → [b]
```

```
scanr f z [] = [z]
scanr f z (x : xs) = f x q : qs
  where qs@(q : _) = scanr f z xs

scanr1 :: (a → a → a) → [a] → [a]
scanr1 f [x] = [x]
scanr1 f (x : xs) = f x q : qs
  where qs@(q : _) = scanr1 f xs
scanr1 _ [] = error "scanr1: empty list"
```

List Generators

```
iterate :: (a → a) → a → [a]
iterate f x = x : iterate f (f x)

repeat :: a → [a]
repeat x = xs where xs = x : xs
```

```
replicate :: Int → a → [a]
replicate n x = take n (repeat x)

cycle :: [a] → [a]
cycle [] = error "cycle: empty list"
cycle xs = xs' where xs' = xs ++ xs'
```

String-Based Functions

```
lines :: String → [String]
lines "" = []
lines s = let (l, s') = break (≡ ' \n') s
  in l : case s' of
    [] → []
    (_ : s'') → lines s''
```

```
words :: String → [String]
words s = case dropWhile Char.isSpace s of
  "" → []
  s' → w : words s''
  where (w, s'') = break Char.isSpace s'
```

```
unlines :: [String] → String
unlines = concatMap (++) "\n"
```

```
unwords :: [String] → String
unwords [] = ""
```

```
unwords ws = foldr1 (\w s → w ++ " " : s) ws
```

```
reverse :: [a] → [a]
reverse = foldl (flip (:)) []
```

Boolean List Functions

```
and, or :: [Bool] → Bool
and      = foldr (∧) True
or       = foldr (∨) False
```

```
any, all :: (a → Bool) → [a] → Bool
any p    = or ∘ map p
all p    = and ∘ map p
```

List Membership Functions

```
elem, notElem :: (Eq a) ⇒ a → [a] → Bool
elem x         = any (≡ x)
notElem x      = all (≠ x)
```

```
lookup :: (Eq a) ⇒ a → [(a, b)] → Maybe b
lookup key [] = Nothing
lookup key ((x, y) : xys)
  | key ≡ x    = Just y
  | otherwise = lookup key xys
```

Arithmetic on Lists

```
sum, product :: (Num a) ⇒ [a] → a
sum           = foldl (+) 0
product      = foldl (*) 1
```

```
maximum, minimum :: (Ord a) ⇒ [a] → a
```

```
maximum [] = error "maximum: empty list"
maximum xs = foldl1 max xs
minimum [] = error "minimum: empty list"
minimum xs = foldl1 min xs
```

List Combining Functions

```
map :: (a → b) → [a] → [b]
map f [] = []
map f (x : xs) = f x : map f xs
```

```
(++) :: [a] → [a] → [a]
[] ++ ys = ys
(x : xs) ++ ys = x : (xs ++ ys)
```

```
concat :: [[a]] → [a]
concat xss = foldr (++) [] xss
```

```
concatMap :: (a → [b]) → [a] → [b]
concatMap f = concat ∘ map f
```

```
zip :: [a] → [b] → [(a, b)]
zip = zipWith (,)
```

```
zip3 :: [a] → [b] → [c] → [(a, b, c)]
zip3 = zipWith3 (,,)
```

```
zipWith :: (a → b → c) → [a] → [b] → [c]
zipWith z (a : as) (b : bs)
  = z a b : zipWith z as bs
zipWith _ _ _ = []
```

```
zipWith3 :: (a → b → c → d) → [a] → [b] → [c] → [d]
zipWith3 z (a : as) (b : bs) (c : cs)
  = z a b c : zipWith3 z as bs cs
zipWith3 _ _ _ _ = []
```

```
unzip :: [(a, b)] → ([a], [b])
unzip = foldr (\(a, b) ~ (as, bs) → (a : as, b : bs)) ([], [])
unzip3 :: [(a, b, c)] → ([a], [b], [c])
unzip3 =
  foldr (\(a, b, c) ~ (as, bs, cs) → (a : as, b : bs, c : cs))
    ([], [], [])
```